

Webページ利用の小さな予約システムの開発

—システム開発の一教材として—

片 山 益 男

A Tiny Reservation System on Web for an Educational Material

KATAYAMA Masuo

要 旨

実用しているweb利用予約システムの開発内容を，教材としての利用を意図して説明した。

Abstract

This article represents a whole process of development of an application using web pages and a server database, intending to serve as an educational material for system development.

1 はじめに

大学学部(文科系)4年生のゼミ指導は大変である。就職活動のために学生の過半数は、少なくとも前期は毎回出席できない。出席を強要すると就職活動に大きな支障が生じる。そのような状況の中で、学生に卒業論文作成への取り組みを持続させるために、日時を予約して個別指導を行うという方法がよく採られる。しかし、各学生の就職活動の進行に伴い、予約変更をしなければならないこともよく生じる。そのための連絡や、予定変更記録の手数も馬鹿にならない。

その労力の軽減をはかるために、Webページ利用による予約という方法を考えついた。予約、取消、変更をすべて学生が入力して、予約状態を自動的に更新していこうというシステムである。Web利用の予約システムは、すでに多くの実用例があるので、方法の解説も豊富であり、雛形をもとに作れば容易に実現できるのではないかと考えていたのであるが、実際に少し調べてみると、そんなに簡単に見つからず、たまたま見付けた事例を読み、予約という業務に汎用的なシステムを作ることが容易ではないことを発見した。

そこで今回の目的に適合したシステムを（老人の趣味として）半月近くの期間をかけて開発したのであるが、その作業量が、学生にとって卒論1つ分に相当するくらいになると判断したので、システム開発の教材として役に立つかも知れないと考え、発表することにした。

以降での説明の仕方は、研究論文という観点から見ると非常に冗長であるが、システム開発の思考過程を説明し、教材としての利用を意図して、このようなスタイルとした。

2 システム仕様の明確化

システムに対する基本的な要件は単純である。どの授業日に指導を予約するかを、学生が web ページから予約する；特定の日には、あまり多くの学生が指導を申し込むと時間的に指導出来なくなるので、一日の指導人数に上限を設ける；後日、学生が予約変更の必要が生じると、学生がやはり web ページから予約変更する；というものである。

まず既存の文献を調査した。そこで見つけた事例¹⁾は、会議室予約のように、特定のマスに誰かが予約するという形式のものであった。今回開発したいのは、あるマスに一定人数まで予約できるというタイプなので、それをそのまま雛形として使うことはできなかった。

しかし予約記録のためには一人につき1つのマス(記録フィールド)を使う必要がある。1予約日(授業日)に予約上限人数分のマスを設けておき、順番に記録していくのが最もシンプルな処理方法である。そしたら、予約変更が生じたとき、どのマスに記録されているかを、どのような方法で指定するのか(探索するのか)を同時に考えなければならない。予約キャンセルで歯抜けになったマスを放置するのか、追加予約があったとき、空いたマスを探して埋めるのか、そうしたら予約先着順という要請があったとき、どのように対処するのも決めておく必要がある。多少の試行錯誤の後、予約は先着順に記録し、キャンセルが出れば、後に並んでいる人を順番に詰めていくという方式をとることにした。それにより追加予約の人を行列の最後に置くことが容易になる。このような方式を決めるに当たっては、配列データの記録位置をどのように変更すればよいかというテクニックを知っていること、または工夫することが必要になる。

次に、利用者が予約指定をするという場面を考える。最小限入力しなければならないのは、何時、誰がというデータである。その時に現在の予約状態が表示されていることが必要になる。

「どの予約日に」を指定する方法として上述の事例に使われていた方法を借用した。予

約日(授業日)のリストから、該当する日をクリックすることにより指定する方法である。それによりキーボードから日付を入力する必要がなくなる。(過去の予約日に対してはアクセス出来なくしておくことも必要である。)

利用者IDの入力データとしては学籍番号を用いることにした。そして、予約状況のリスト表示には氏名を用いた。予約リストは誰でも見ることができるので、そこに利用者ID(学籍番号)を表示すれば、ちょっとしたいたずらで、予約変更などが他人により容易に実行されてしまうからである。厳格にしようとするれば、パスワードの入力も必要になるが、今回は利用者IDだけを用いる略式保護とした。当然、利用可能者のリストを事前に準備しておき、それ以外の人々の入力を許可しないという処理が必要となる。

予約取消という処理も、新規予約と同様の処理方法を取ればよい。予約内容の変更は、処理プログラムを単純にするため、「取消」と「新規予約」の2つの処理を利用者に実行してもらうこととした。そこで、利用者が「取消」と「新規予約」のどちらかを選択する方法を準備しておく必要がある。それぞれ別のページを準備しておき、利用者に処理ページを選ばせるというのも一つの方法である。ここでは、処理対象とする予約日をまず選択し、次に「取消」か「新規予約」か、を選択するという方式を用いた。その時、もし予約が満杯であれば、新規予約の項目は表示させないという工夫により、誤入力の可能性を少なくした。

「取消」または「新規予約」の処理をすれば、変更後の予約リストを自動的に表示させるのが当然であるが、用いた処理系(IISとASP)で、その方法がわからず、「確認」ボタンを設けて変更後の予約リストを表示させるようにしている。

以上で、利用者(学生)のみで運用できる最低限の予約システムは実行可能になるが、もう少し用途を広げたいという欲が出てくる。例えば、予約しても出席しなかった学生は目立つように表示したい、逆に学生側からは、教員側の都合で、途中で時間切れになり未指導者が出た場合、休講で指導できなかつた場合も表示すべきだ、予約していなくとも当日出席して指導した学生も表示したい、などである。それらを、表示色を変えて表示するような仕組みを追加した。

具体的には、予約外処理を行うためのページを準備し、該当予約日→利用者ID→予約外処理の種類(欠席、追加指導、教員側理由による未指導)を順番に指定すると、記録している利用者IDの前に記号が追加されるようにし、その記号を用いて、予約一覧リストで氏名の色が変化するようにした。この部分を追加することにより、単なる指導予約だけでなく、過去の指導状況の一覧表示も可能になった。

システム開発という立場から補足しておきたいのは、このようなシステム仕様は、開発

の最初から明確であったわけではなく、開発進行と共に明確化が進んでいくという特徴であり、開発には試行錯誤的なプロセスを伴うことを強調しておきたい。それをすべて否定するような進め方では、使いやすいシステムとならないことが多いと考えている。

3 記録すべきデータ

この予約システムを使っていくために最小限必要なデータは、利用者のリストと、予約状態の2種類である。ここではMS-Accessを用い、前者を利用者テーブル、後者を予約テーブルとして作成した。

利用者テーブルには利用者ID(学籍番号)と氏名、の2項目を設けた。利用者が予約データを入力するとき、登録済みかどうかをチェックするため、及び予約リストに予約者の氏名を表示するために用いる。

予約テーブルの項目は、行番号、予約日の日付、予約者の利用者IDリスト、対応する氏名のリスト、予約者数、の5項目である。各予約日には複数の利用者の予約が可能なので、本来は複数の欄を準備しておくべきであるが、join, split という関数を用いることにより、複数利用者のIDや氏名を1つの文字列として記録することができる。

実際の指導が予約と異なった場合、色を変えて表示するという工夫について、具体的には利用者IDの前に次のような記号をつけることで、予約外の処理を記録することとした。

欠席：	*
追加指導：	@
教員都合による未指導：	\$

(当初は、追加指導に対しては「+」の記号をつけるように設定したが、この記号はASPで文字処理をするとき、予約記号として扱われ、正しい処理が行えないので変更した。)

その外、利用者が個々に予約や取消処理をした履歴を記録するための予約ログ・テーブル、教員による予約外処理の履歴を記録するための予約外処理ログ・テーブルも準備した。

これら4つのテーブルを含むデータベース・ファイルをyoyaku.mdbという名前で作成した。各テーブルの名称はgaku, sts, yoy, sidとした(これらの名称はASPファイルの処理内容を解読するとき必要となる)。

利用に先立って、利用者テーブルに利用者IDと氏名データ、予約テーブルに予約日(授業日)データを入力しておく必要がある。予約テーブルの予約者数はゼロ、利用者IDリストと氏名リストの項目には、予約上限数だけの斜線(join, splitで用いる区切文字)を初期値として設定しておく。以上のテーブルと初期値を図1に示す。

studentid	lstname	fstname
g01	青木	
g02	木下	
g03	久下	
g04	小西	
g05	佐藤一	
g06	佐藤新	
g07	田中	
g08	中村	
g09	吉田	

no	opendate	resvno	ylist	ynlist
1	2005/06/20	0	/////	/////
2	2005/06/21	0	/////	/////
3	2005/06/23	0	/////	/////
4	2005/06/24	0	/////	/////
5	2005/06/25	0	/////	/////
6	2005/06/26	0	/////	/////
7	2005/06/27	0	/////	/////
8	2005/06/28	0	/////	/////
9	2005/06/29	0	/////	/////
10	2005/07/01	0	/////	/////
11	2005/07/02	0	/////	/////
12	2005/07/05	0	/////	/////
0		0		

seq	opendate	studtid	reccdate	type
(オー				

seq	opendate2	studid	typ2	reccdate
(オー				

図1 4種のテーブルと、その初期値

左上：利用者テーブル (gaku), 右上：予約テーブル (sts)

左下：予約ログ (yoy), 右下：予約外処理ログ (sid)

4. システムの利用

この章では、利用手順と、利用のための画面操作により、どのような処理が行われるのかを関連づけながら説明する。

4.1 予約リストの表示

学生が予約、またはその取消のために、最初のページ (y.asp) にアクセスすると予約状況の一覧リストが表示される (図2)。

Web サーバーが y.asp へのアクセスを受けたとき、html 文書として送り出すための処理内容が y.asp に記されているのであるが (リスト1)、ここでの主要な処理は、予約テーブルの内容を表形式の html 文書に変換することである。

まずデータベース・ファイル (yoyaku.mdb) に接続し、予約テーブルを読み取り専用モードでオープンする。そして、その各行を順番に読み取りながら、その内容をhtml 文書の表データに変換していくのであるが、その変換作業がかなり複雑になる。

最初の行番号のところへは、その部分をクリックすれば当該予約日への予約・予約取消が行えるページ (y1.asp) へのハイパーリンクを設定している。単にy1.aspへのハイパーリンクを設定するだけなら単純であるが、どの予約日を対象とするのかというデータを同

予約一覧

#	予約日	予	約	者	
1	2005/06/20	田中	久下	小西	佐藤新
2	2005/06/21	佐藤一	佐藤新	田中	
3	2005/06/23	青木	久下	佐藤一	佐藤新 田中
4	2005/06/24	木下			
5	2005/06/25	中村			
6	2005/06/26				
7	2005/06/27	吉田	木下	久下	吉田
8	2005/06/28				
9	2005/06/29				
10	2005/07/01				
11	2005/07/02				
12	2005/07/05				

予約の追加・削除をしたい行の番号をクリックしてください。
(過去の日付はクリックできません。)

予約日を変更したい時は「追加」と「削除」の、2回の処理を行ってください。

当日になってからの予約変更(追加・削除)はできません。

過去の予約日について、

- ・文字が赤色の名前は、予約日に欠席して、指導しなかった人です。
- ・文字がオレンジ色の名前は、時間切れ、休講などにより、予約日に指導できなかった人です。
- ・文字が青色の名前は、予約なしに、予約日に指導した人です。
- ・文字が黒色の名前は、予約通り、予約日に指導した人です。

図2 y.aspの実行により表示される画面

時に添付する必要がある(サーバーへのget形式データ送信)。例えばサンプル・データ(図1)では1行目の予約日は“2005/06/20”となっているので、html文書での予約リスト1行目、行番号項目のデータ内容を次のように指定する必要がある:

```
<a ref=" y1.asp?datum=2005/06/20" >1 </a>
```

行番号と予約日の数値は行毎に変化するのので、予約テーブルに記録しているデータを読み取り、その内容をハイパーリンク指定の部分へ埋め込む処理を、繰り返し行えるようなスクリプトをaspファイルに記述しておく必要がある。それがリスト1では次の部分になる。

```
<a href="y1.asp?keyid=<% =repRec("no") %>&rno=<% =repRec("resvno") %>&datum=<% =theday %>">
```

```
<% = repRec("no") %> </a>
```

(このスクリプトではy1.aspにアクセスするとき、予約日データの他に、行番号、予約者数のデータも同時に添付送信している。これらのデータはy1.aspですぐに使うのではなく、

実際に予約テーブルの内容を更新するとき用いる。html ページはそれぞれ独立しているため、ページ間の連携をとるために、このような工夫が必要となる。

さらに、予約日指定は、過去の日付に対して行うことはできないので、その部分へのハイパーリンクは設定しないようにしておかなければならない。

次に、各行につき、予約があれば、その氏名を右に表示していく。上述のように、データそのものは一つの文字列として予約テーブルに記録されているので、それを読み出し、メモリー内で配列データとして分離し、html 文書の表の各欄に埋め込んでいくという処理を行う。それがリスト 1 では次の部分となる (予約者数上限を 7 人としている)。

```
<% pc = split(repRec("ynlist"), "/")
for i = 0 to 6 %>
  <TD BORDERCOLOR=#c0c0c0 ><%= pc(i) %><BR></TD>
<% next %>
```

それに加え、過去の予約日については、欠席、追加指導などの氏名を色分け表示するようにしたので、その部分の処理が追加されている。すなわち予約者IDデータの前に記号が付いている場合、表示する氏名データに表示色を指定する処理を追加している。図 2 に対応する予約テーブルの内容を図 3 に例示する。

sts : テーブル					
	no	opendate	resvno	ylst	ynlist
▶	1	2005/06/20	4	@g07/*g03/@g04/@g06///	田中/久下/小西/佐藤新///
	2	2005/06/21	3	\$g05/*g06/@g07///	佐藤一/佐藤新/田中///
	3	2005/06/23	5	g01/g03/\$g05/g06/g07///	青木/久下/佐藤一/佐藤新/田中//
	4	2005/06/24	1	g02/////	木下/////
	5	2005/06/25	1	g08/////	中村/////
	6	2005/06/26	0	/////	/////
	7	2005/06/27	4	g09/g02/g03/g09///	吉田/木下/久下/吉田///
	8	2005/06/28	0	/////	/////
	9	2005/06/29	0	/////	/////
	10	2005/07/01	0	/////	/////
	11	2005/07/02	0	/////	/////
	12	2005/07/05	0	/////	/////
*	0		0		

図 3 図 2 に対応する予約テーブルの内容

最初から、このような複雑なスクリプトを作成することは困難で、エラーが生じたとき、どのように対処すればよいのか分からなくなる。まず単純な一覧リスト表示ができるスクリプトを作り、それに追加処理を逐次加えていく行き方が実際的である。エラーが発生したとき、web ページに表示される内容だけでは原因がつかめない場合も多い。その時、システム・ファイル内にある、web ページへのアクセス・ログ・ファイルを見ると、もう少し具体的な内容が記載されていることが多いので利用を勧める。その位置は、サーバーが

インストールされているドライブの、winnt\system32\logfiles\w3svc1\である。

4.2 利用者IDの入力と処理種類の選択

予約リストから、処理したい授業日の行をクリックすると、y1.asp (リスト2) が実行され、利用者IDなどの入力画面が表示される (図4)。

a) 予約追加・削除日: 2005/06/28
予約追加 ID

b) 予約追加・削除日: 2005/06/27
処理したい行だけに ID を入力してください。
予約取消 ID
予約追加 ID

c) 予約追加・削除日: 2005/06/27
予約取消 ID

図4 3種の画面のどれか1つが表示される。

このとき、既存の予約者数に応じて3種の画面の一つが表示され、誤入力の軽減を計るようになっている。まず予約者が0のときは、新規予約(追加予約)のみが可能な画面が表示される。逆に、予約上限まで予約が入っているときは、予約取消のみが可能な画面となる。その中間では、両方の入力ができるような画面となり、その場合は、どちらか一方だけに利用者IDを入力して、そのボタンを押すという操作方式を取っている。

処理種類のボタンを選択して押すと、y.aspから受け取ったデータに、利用者IDと処理種類のデータが追加され、post形式でそれらのデータがサーバーに送られ、同時にy2.aspが実行される。

4.3 入力データの確認

y2.asp (リスト 3) が実行されると、まず入力された利用者IDが利用者テーブルに登録されているかどうかをチェックする。登録されていれば、それまでの入力データが表示され、予約テーブルの更新を実行してもよいかどうかの確認が求められる (図 5)。確認ボタンを押すと、必要なデータがpost形式でサーバーに送られ、y3.aspが実行される。キャンセル・ボタンで処理を中断することもできる。

#	予約日	予	約	者		
1	2005/06/20	田中	久下	小西	佐藤新	
2	2005/06/21	佐藤一	佐藤新	田中		
3	2005/06/23	青木	久下	佐藤一	佐藤新	田中
4	2005/06/24	木下				
5	2005/06/25	中村				
6	2005/06/26					
7	2005/06/27	吉田	木下	久下	吉田	
8	2005/06/28					
9	2005/06/29					
10	2005/07/01					
11	2005/07/02					
12	2005/07/05					

次の処理を実行します。よろしいですか？

日付:2005/06/28

ID : g08

処理: 予約追加

確認

キャンセル

図 5 予約・取消 処理実行前の確認画面

4.4 予約テーブルの更新

y3.asp (リスト 4) の実行により、予約テーブルの該当行の内容が読み込まれ、新規予約の場合は、予約者リストに利用者IDと氏名データが追加されると同時に予約者数が1つ増える。

予約取消の場合は、まず予約者リストに、指定された利用者IDが記録されているかどうかを調べ、記録されているときは、その利用者データを削除すると共に、後ろに並んでいる予約者の位置を前に詰め、予約者数を1つ減らす。そして変更内容を予約テーブルに書き込む。さらに、予約ログ・テーブルに、そのトランザクションを追加する。

当然のことながら、予約テーブルの更新結果が自動的に表示されるべきであるが、その方法が不明なため（ASPにはserver transferという処理も準備されているのであるが、それを使ったのでは更新結果を表示させることができないので、筆者の不勉強のためか?）、y.aspへのハイパーリンクを準備するという手段でお茶を濁している。

予約内容を変更しました。

[新しい予約一覧を見る](#)

図6 処理実行後の画面

4.5 予約日（授業日）終了後の予約外処理

欠席、追加指導、指導未完了などを追加記録するためには、s.asp（リスト5）を実行する。ここで表示されるのは過去の予約日（授業日）の部分だけである。

それ以降の処理手順の説明は割愛し、画面（図7～10）とリスト（リスト6～8）の紹介にとどめる。

予約一覧

#	予約日	予	約	者		
1	2005/06/20	田中	久下	小西	佐藤新	
2	2005/06/21	佐藤一	佐藤新			
3	2005/06/23	青木	久下	佐藤一	佐藤新	田中
4	2005/06/24	木下				
5	2005/06/25	中村				
6	2005/06/26					

予約外指導、欠席を記録したい行の番号をクリック

図7 s.asp 実行により表示される画面

Webページ利用の小さな予約システムの開発 (片山益男)

予約日: 2005/06/23 予約: g01/g03/g05/g06/g07//

ID	氏名
g01	青木
g02	木下
g03	久下
g04	小西
g05	佐藤一
g06	佐藤新
g07	田中
g08	中村
g09	吉田

当日指導または欠席者の ID をクリック

図 8 s1.asp 実行 (予約日のクリック) により表示される画面

予約日: 2005/06/23 予約: g01/g03/g05/g06/g07//

ID : g05

追加指導

学校理由による未指導

欠席

キャンセル

図 9 s2.asp 実行 (ID のクリック) により表示される画面

2005/06/21

g07

田中

add

当日指導などを記録しました。

同じ予約日の記録を追加・変更する

修正結果を見る

図10 s3.asp 実行 (追加指導, 欠席などのボタンを押すこと) により表示される画面

5 おわりに

実際に使おうという動機から開発し、その後、教材として使えるのではないかと気がついて発表することにした。(このシステムは実際に利用しており、卒論完成への自主的な取り組みが、従来より少し高まったような気がする。)

今一つ、このような小さなシステムを気軽に使える環境(このようなフリー・ソフトの充実、各教員が自分のサーバーが使える環境など)の整備が、教育へのIT利用拡大に必要ではないかという思いがある。

なお、この開発には過去の経験を利用している^{2,3)}。

参考文献

1. 西沢直木, 「ASPによるWebアプリケーション・スーパーサンプル」, pp.295-307, ソフトバンク, 2001
2. 拙稿, 「Web利用自主学习システムの一構成モデル」, 大阪産業大学経営論集, 第2巻, 第2号, pp.15-49, 2001
3. 拙稿, 「Webページを用いた教育用ゲーム開発事例」, 大阪産業大学経営論集, 第5巻, 第3号, pp.17-39, 2004

リスト 1 y.asp

```

<%@ LANGUAGE="VBScript" %>
<HTML>
<HEAD>
<TITLE>予約一覧</TITLE>
</HEAD>
<meta http-equiv="Content-Type" content="text/html;charset=Shift_JIS">
<body>
<div align="center">
<br>
<P>予約一覧</P>
<%
    Response.Expires = -1
    Set repConn = Server.CreateObject("ADODB.Connection")
    Set repRec = Server.CreateObject("ADODB.Recordset")
    repConn.Open "yoyaku.mdb"
    strSQL = "select * from sts order by opendate"
    repRec.Open strSQL, repConn, 3, 1, 1
%>
<TABLE BORDER=1 BGCOLOR=#ffffff CELLSPACING=0>
<THEAD>
<TR>
<TH BGCOLOR=#c0c0c0 BORDERCOLOR=#000000 >#</TH>
<TH BGCOLOR=#c0c0c0 BORDERCOLOR=#000000 >予約日</TH>
<TH BGCOLOR=#c0c0c0 BORDERCOLOR=#000000 >予</TH>
<TH BGCOLOR=#c0c0c0 BORDERCOLOR=#000000 >約</TH>
<TH BGCOLOR=#c0c0c0 BORDERCOLOR=#000000 >者</TH>
</TR>
</THEAD>
<TBODY>
<% Do Until repRec.EOF
    theday = repRec("opendate") %>
<TR VALIGN=TOP>
<TD BORDERCOLOR=#c0c0c0 >
<% if theday > date then %>
    <a href="y1.asp?keyid=<% =repRec("no")
%>&rno=<% =repRec("resvno") %>&datum=<% =theday %>">
    <%= repRec("no") %> </a>
<% else %>

```

```

    <%= repRec("no") %>
<% end if %>
    <BR></TD>
<TD BORDERCOLOR=#c0c0c0 ><%= theday %><BR></TD>
<%
    pc = split(repRec("ynlist"), "/")
    pc1 = split(repRec("ylist"), "/")
    for i = 0 to 6
    if (theday < date) and (Left(pc1(i),1) = "*") then %>
        <TD BORDERCOLOR=#c0c0c0 ><font color="red">
<%= pc(i) %></font><BR></TD>
    <% elseif (theday < date) and (Left(pc1(i),1) = "@") then %>
        <TD BORDERCOLOR=#c0c0c0 ><font color="blue">
<%= pc(i) %></font><BR></TD>
    <% elseif (theday < date) and (Left(pc1(i),1) = "$") then %>
        <TD BORDERCOLOR=#c0c0c0 ><font color="orange">
<%= pc(i) %></font><BR></TD>
    <% else %>
        <TD BORDERCOLOR=#c0c0c0 ><%= pc(i) %><BR></TD>
    <% end if %>
    <% next %>
</TR>
<%
    repRec.MoveNext
    loop
%>
<%
    repRec.Close
    repConn.Close
%>
</TBODY>
</TABLE>
</div>
<br><br>
    予約の追加・削除をしたい行の番号をクリックしてください。<br>
    (過去の日付はクリックできません。) <br><br>
    予約日を変更したい時は「追加」と「削除」の、2回の処理を行ってください。
<br><br>
    当日になってからの予約変更（追加・削除）はできません。<br><br>
    過去の予約日について、<br><br>
    ・文字が<font color="red">赤色</font>の名前は、予約日に欠席して、

```

指導しなかった人です。

・文字がオレンジ色の名前は, 時間切れ, 休講などにより, 予約日に指導できなかった人です。

・文字が青色の名前は, 予約なしに, 予約日に指導した人です。

・文字が黒色の名前は, 予約通り, 予約日に指導した人です。

</BODY>

</HTML>

リスト 2 y1.asp

<HTML>

<HEAD>

<TITLE>予約追加・削除</TITLE>

</HEAD>

<meta http-equiv="Content-Type" content="text/html;charset=Shift_JIS">

<%

kid = Request("keyid")

rsno = Request("rno")

opendate = Request("datum")

%>

<body>

予約追加・削除日 : <% =opendate %>

<% if rsno > 0 and rsno < 6 then %>

処理したい行だけに ID を入力してください。

<% end if %>

<% if rsno > 0 then %>

<form method="POST" action="y2.asp">

<input type="hidden" name="keyid" value="<% =kid %>">

<input type="hidden" name="datum" value="<% =opendate %>">

<input type="hidden" name="typ" value="delt">

<input type="Submit" name="sub" value="予約取消"> ID

<input type="text" name="sid" value="<% =sid %>">

</form>

<% end if %>

<% if rsno < 6 then %>

<form method="POST" action="y2.asp">

<input type="hidden" name="keyid" value="<% =kid %>">

<input type="hidden" name="datum" value="<% =opendate %>">

<input type="hidden" name="typ" value="add">

```

<input type="Submit" name="sub" value="予約追加"> ID
<input type="text" name="sid" value="<% =sid %>"><br><br>
</form>
<% end if %>
</body>
</html>

```

リスト 3 y2.asp

```

<HTML>
<HEAD>
<TITLE>予約追加・削除2</TITLE>
</HEAD>
<meta http-equiv="Content-Type" content="text/html;charset=Shift_JIS">
<body>
<%
    dim kid, opendate, typ, stid, lname
    kid = Request("keyid")
    opendate = Request("datum")
    typ = Request("typ")
    stid = LCase(Request("sid"))
    Set repConn = Server.CreateObject("ADODB.Connection")
    Set repRec = Server.CreateObject("ADODB.Recordset")
    repConn.Open "yoyaku.mdb"
    strSQL = "select * from gaku where studentid = " & stid & ";"
    repRec.Open strSQL, repConn, 3, 1, 1
    if repRec.EOF then
%>
        入力した ID は登録されていません。「戻る」ボタンで再入力してください。<br>
<% else
    lname = repRec("lstname")
    call updconf
    end if
    repRec.Close
    repConn.Close
%>
<%
Sub updconf() %>
<%
    Set repRec2 = Server.CreateObject("ADODB.Recordset")

```



```

strSQL2 = "select * from sts order by opendate"
repRec2.Open strSQL2, repConn, 3, 1, 1
%>
<div align="center">
<TABLE BORDER=1 BGCOLOR=#ffffff CELLSPACING=0>
<THEAD>
<TR>
<TH BGCOLOR=#c0c0c0 BORDERCOLOR=#000000 >#</TH>
<TH BGCOLOR=#c0c0c0 BORDERCOLOR=#000000 >予約日</TH>
<TH BGCOLOR=#c0c0c0 BORDERCOLOR=#000000 >予</TH>
<TH BGCOLOR=#c0c0c0 BORDERCOLOR=#000000 >約</TH>
<TH BGCOLOR=#c0c0c0 BORDERCOLOR=#000000 >者</TH>
</TR>
</THEAD>
<TBODY>
<% Do Until repRec2.EOF %>
<TR VALIGN=TOP>
<TD BORDERCOLOR=#c0c0c0 ><%= repRec2("no") %><BR></TD>
<TD BORDERCOLOR=#c0c0c0 ><%= repRec2("opendate") %><BR></TD>
<%
pc = split(repRec2("ynlist"), "/")
for i = 0 to 6
%>
<TD BORDERCOLOR=#c0c0c0 ><%= pc(i) %><BR></TD>
<%
next
%>
</TR>
<% repRec2.MoveNext
loop
repRec2.Close
%>
</TBODY>
</TABLE>
</div>
<br><br>
次の処理を実行します。よろしいですか? <br><br>
日付 : <%= opendate %><br><br>
ID : <%= stid %><br><br>
処理 :
<% if typ = "delt" then %>

```

```

予約取消
<% else %>
予約追加
<% end if %>
<br>
<form method="POST" action="y3.asp">
<input type="hidden" name="keyid" value="<% =kid %>">
<input type="hidden" name="datum" value="<% =opendate %>">
<input type="hidden" name="typ" value="<% =typ %>">
<input type="hidden" name="sid" value="<% =stid %>">
<input type="hidden" name="lname" value="<% =lname %>">
<input type="Submit" name="sub" value="確認">
</form>
<form method="POST" action="y.asp">
<input type="Submit" name="sub" value="キャンセル">
<br>
<%
End Sub
%>
</body>
</html>

```

リスト4 y3.asp

```

<HTML>
<HEAD>
<TITLE>予約内容の変更</TITLE>
</HEAD>
<meta http-equiv="Content-Type" content="text/html;charset=Shift_JIS">
<body>
<%
dim kid, opendt, typ, stid, lname
kid = Request("keyid")
opendt = Request("datum")
typ = Request("typ")
stid = Request("sid")
lname = Request("lname")
Set repConn = Server.CreateObject("ADODB.Connection")
Set repRec = Server.CreateObject("ADODB.Recordset")
repConn.Open "yoyaku.mdb"

```

```

StrSQL = "select * from sts where opendate = #" & opendt & "#;"
Application.Lock
repRec.Open StrSQL, repConn, 3, 3, 1
%>
<% if repRec.EOF then %>
error. EOF desu.<br><br>
<% end if %>
<%
ysu = repRec("resvno")
ary1 = split(repRec("ylist"), "/")
ary2 = split(repRec("ynlist"), "/")
if typ = "delt" then
    call deletep
else
    call addp
end if
repRec("ylist") = join(ary1, "/")
repRec("ynlist") = join(ary2, "/")
repRec("resvno") = ysu
repRec.update
repRec.Close
Application.Unlock
Application.Lock
Set repRec2 = Server.CreateObject("ADODB.Recordset")
repRec2.Open "yoy", repConn, 3, 3, 2
repRec2.AddNew
repRec2("opendate") = opendt
repRec2("studtid") = stid
repRec2("recdate") = Date
repRec2("type") = typ
repRec2.Update
repRec2.Close
Application.Unlock
repConn.Close
%>
<br>予約内容を変更しました。<br>
<br><a href="y.asp">新しい予約一覧を見る</a><br>
<% Sub deletep()
sw = 0
for i = 0 to ysu-1
if ary2(i) = lname then

```

```

ary2(i) = ""
ary1(i) = ""
sw = 1
if i < ysu-1 then
  for j = i to ysu-1
    ary2(j) = ary2(j+1)
    ary1(j) = ary1(j+1)
    ary2(j+1) = ""
    ary1(j+1) = ""
  next
end if
exit for
end if
next
if sw = 0 then
%>
<br> 指定した日に、該当する ID の予約はありませんでした。<br>
<br>
<% else
  ysu = ysu - 1
  end if
  End Sub %>
<% Sub addp()
  ary1(ysu) = stid
  ary2(ysu) = lname
  ysu = ysu + 1
  End Sub %>
</body>
</html>

```

リスト5 s.asp

```

<%@ LANGUAGE="VBScript" %>
<HTML>
<HEAD>
<TITLE>例外指導記録1</TITLE>
</HEAD>
<meta http-equiv="Content-Type" content="text/html;charset=Shift_JIS">
<body>
<div align="center">

```

```

<br>
<P>予約一覧</P>
<%
  Response.Expires = -1
  Set repConn = Server.CreateObject("ADODB.Connection")
  Set repRec = Server.CreateObject("ADODB.Recordset")
  repConn.Open "yoyaku.mdb"
  strSQL = "select * from sts order by opendate"
  repRec.Open strSQL, repConn, 3, 1, 1
%>
<TABLE BORDER=1 BGCOLOR=#ffffff CELLSPACING=0>
<THEAD>
<TR>
<TH BGCOLOR=#c0c0c0 BORDERCOLOR=#000000 >#</TH>
<TH BGCOLOR=#c0c0c0 BORDERCOLOR=#000000 >予約日</TH>
<TH BGCOLOR=#c0c0c0 BORDERCOLOR=#000000 >予</TH>
<TH BGCOLOR=#c0c0c0 BORDERCOLOR=#000000 >約</TH>
<TH BGCOLOR=#c0c0c0 BORDERCOLOR=#000000 >者</TH>
</TR>
</THEAD>
<TBODY>
<% Do Until repRec("opendate")>Date
  theday = repRec("opendate") %>
<TR VALIGN=TOP>
<TD BORDERCOLOR=#c0c0c0 >
  <a href="s1.asp?ylist=<% =repRec("ylist")
  %>&datum=<% =theday %>">
  <%= repRec("no") %> </a>
  <BR></TD>
<TD BORDERCOLOR=#c0c0c0 ><%= theday %><BR></TD>
<%
  pc = split(repRec("ynlist"), "/")
  pc1 = split(repRec("ylist"), "/")
  for i = 0 to 6
  if (theday < date) and (Left(pc1(i),1) = "*" ) then %>
    <TD BORDERCOLOR=#c0c0c0 ><font color="red">
<%= pc(i) %></font><BR></TD>
<% elseif (theday < date) and (Left(pc1(i),1) = "@" ) then %>
    <TD BORDERCOLOR=#c0c0c0 ><font color="blue">
<%= pc(i) %></font><BR></TD>
<% elseif (theday < date) and (Left(pc1(i),1) = "$" ) then %>

```

```

    <TD BORDERCOLOR=#c0c0c0 ><font color="orange">
<%= pc(i) %></font><BR></TD>
<% else %>
    <TD BORDERCOLOR=#c0c0c0 ><%= pc(i) %><BR></TD>
<% end if %>
<% next %>
</TR>
<%
    repRec.MoveNext
    loop
    repRec.Close
    repConn.Close
%>
</TBODY>
</TABLE>
</div>
<br><br>
    予約外指導, 欠席を記録したい行の番号をクリック<br>
</BODY>
</HTML>

```

リスト6 s1.asp

```

<HTML>
<HEAD>
<TITLE>例外指導記録2</TITLE>
</HEAD>
<meta http-equiv="Content-Type" content="text/html;charset=Shift_JIS">
<%
ylst = Request("ylist")
opendate = Request("datum")
%>
<body>
    予約日 : <%=opendate %> 予約 : <%=ylst %><br>
<div align="center">
<br>
<%
    Set repConn = Server.CreateObject("ADODB.Connection")
    Set repRec = Server.CreateObject("ADODB.Recordset")
    repConn.Open "yoyaku.mdb"

```

```
StrSQL = "select * from gaku order by studentid"
repRec.Open StrSQL, repConn, 3, 1, 1
%>
<TABLE BORDER=1 BGCOLOR=#ffffff CELLSPACING=0>
<THEAD>
<TR>
<TH BGCOLOR=#c0c0c0 BORDERCOLOR=#000000 > ID </TH>
<TH BGCOLOR=#c0c0c0 BORDERCOLOR=#000000 >氏名</TH>
</TR>
</THEAD>
<TBODY>
<% Do Until repRec.EOF
sid = repRec("studentid") %>
<TR VALIGN=TOP>
<TD BORDERCOLOR=#c0c0c0 >
  <a href="s2.asp?stid=<% =sid %>&lnm=<% =repRec("lstname")
  %>&date2=<% =opendate %>&ylist=<% =ylst %>">
  <%= sid %></a>
  <BR></TD>
<TD BORDERCOLOR=#c0c0c0 ><% = repRec("lstname") %><BR></TD>
</TR>
<%
  repRec.MoveNext
  loop
  repRec.Close
  repConn.Close
%>
</TBODY>
</TABLE>
</div>
<br> 当日指導または欠席者の ID をクリック<br>
</body>
</html>
```

リスト 7 s2.asp

```
<HTML>
<HEAD>
<TITLE>例外指導記録3</TITLE>
</HEAD>
```

```
<meta http-equiv="Content-Type" content="text/html;charset=Shift_JIS">
<%
ylst = Request("ylist")
opendate = Request("date2")
sid = Request("stid")
lname = Request("lnm")
%>
<body>
    予約日 : <% =opendate %>   予約 :   <% =ylist %><br>
<br>
    ID :   <% =sid %><br>
<form method="POST" action="s3.asp">
<input type="hidden" name="stid" value="<% =sid %>">
<input type="hidden" name="lnm" value="<% =lname %>">
<input type="hidden" name="datum" value="<% =opendate %>">
<input type="hidden" name="ylist" value="<% =ylist %>">
<input type="hidden" name="typ" value="add">
<input type="Submit" name="sub" value="追加指導">
</form>
<form method="POST" action="s3.asp">
<input type="hidden" name="stid" value="<% =sid %>">
<input type="hidden" name="lnm" value="<% =lname %>">
<input type="hidden" name="ylist" value="<% =ylist %>">
<input type="hidden" name="datum" value="<% =opendate %>">
<input type="hidden" name="typ" value="not">
<input type="Submit" name="sub" value="学校理由による未指導">
</form>
<form method="POST" action="s3.asp">
<input type="hidden" name="stid" value="<% =sid %>">
<input type="hidden" name="lnm" value="<% =lname %>">
<input type="hidden" name="ylist" value="<% =ylist %>">
<input type="hidden" name="datum" value="<% =opendate %>">
<input type="hidden" name="typ" value="absent">
<input type="Submit" name="sub" value="欠席">
</form>
<form method="POST" action="s1.asp">
<input type="hidden" name="ylist" value="<% =ylist %>">
<input type="hidden" name="datum" value="<% =opendate %>">
<input type="Submit" name="sub" value="キャンセル">
</form>
</body>
```


</html>

リスト 8 s3.asp

```
<%@ LANGUAGE="VBScript" %>
<HTML>
<HEAD>
<TITLE>例外指導記録4</TITLE>
</HEAD>
<meta http-equiv="Content-Type" content="text/html;charset=Shift_JIS">
<body>
<%
dim sw
sw = 0
sid = Request("stid")
ylst = Request("ylist")
lname = Request("lnm")
opendt = Request("datum")
type1 = Request("typ") %>
<% =opendt %><br>
<% =sid %><br>
<% =lname %><br>
<% =type1 %><br>
<%
Set repConn = Server.CreateObject("ADODB.Connection")
Set repRec = Server.CreateObject("ADODB.Recordset")
repConn.Open "yoyaku.mdb"
Application.Lock
StrSQL = "select * from sts where opendate = #" & opendt & "#;"
repRec.Open StrSQL, repConn, 3, 3, 1
%>
<% if repRec.EOF then %>
error. EOF desu.<br><br>
<% end if %>
<%
ysu = repRec("resvno")
ary = split(repRec("ylist"), "/")
ary2 = split(repRec("ynlist"), "/")
if type1 = "absent" then
call absentp
```

```
elseif type1 = "not" then
  call notp
else
  call addp
end if
if sw = 1 then
  repRec("ylist") = join(ary, "/")
  repRec("ynlist") = join(ary2, "/")
  repRec("resvno") = ysu
  repRec.update
  ylst = repRec("ylist")
end if
repRec.Close
Application.Unlock
%>
<%
if sw = 1 then
  Application.Lock
  Set repRec2 = Server.CreateObject("ADODB.Recordset")
  repRec2.Open "sid", repConn, 3, 3, 2
  repRec2.AddNew
  repRec2("opendate2") = opendt
  repRec2("studid") = sid
  repRec2("typ2") = type1
  repRec2("reccdate") = Date
  repRec2.Update
  repRec2.Close
  Application.Unlock
end if
repConn.Close
%>
<br>当日指導などを記録しました。 <br><br>
<form method="POST" action="s1.asp">
<input type="hidden" name="ylist" value="<% =ylst %>">
<input type="hidden" name="datum" value="<% =opendt %>">
<input type="Submit" name="sub" value="同じ予約日の記録を追加・変更する">
</form>
<form method="POST" action="s.asp">
<input type="Submit" name="sub" value="修正結果を見る">
</form>
<br><a href="s.asp"></a><br>
```

```
<% Sub absentp()
for i = 0 to ysu-1
if ary(i) = sid then
  ary(i) = "*" & sid
  sw = 1
  exit for
end if
next
if sw = 0 then
%>
<br>  指定した日に、該当する ID の予約はありませんでした。<br>
<br>
<% end if
End Sub %>
<% Sub notp()
for i = 0 to ysu-1
if ary(i) = sid then
  ary(i) = "$" & sid
  sw = 1
  exit for
end if
next
if sw = 0 then
%>
<br>  指定した日に、該当する ID の予約はありませんでした。<br>
<br>
<% end if
End Sub %>
<% Sub addp()
sw = 1
ary(ysu) = "@" & sid
ary2(ysu) = lname
ysu = ysu + 1
End Sub %>
</BODY>
</HTML>
```